Appendix A.1: Introduction to EES

EES (pronounced 'ease') is an acronym for Engineering Equation Solver. The basic function provided by EES is the numerical solution of non-linear algebraic and differential equations. EES is an equation-solver, rather than a programming language, since it does not require the user to enter instructions for iteratively solving non-linear equations. EES provides capability for unit checking of equations, parametric studies, optimization, uncertainty analyses, and high-quality plots. It provides array variables that can be used in finite-difference calculations. In addition, EES provides high-accuracy thermodynamic and transport property functions for many fluids and solid materials that can be integrated with the equations. The combination of all of these capabilities in one program makes EES a very powerful tool for solving heat transfer problems.

A limited academic version of EES that includes all of the example problems from this text can be downloaded from www.cambridge.org/nellisandklein. A commercial or professional version or academic site license of EES can be obtained from:
F-Chart Software
http://fchart.com
email: info@fChart.com

A.1.1 Getting Started

To start EES, double-click on the EES program icon (shown above) or on any file created by EES and having the *.ees filename extension. You can also start EES from the Windows Run command in the Start menu by entering EES and clicking the OK button. EES begins by displaying a dialog window, which shows registration information, the version number and other information. Click the OK button to dismiss the dialog window.

Detailed help is available at any time while working with EES; pressing the F1 key will bring up a Help window relating to the foremost window. Clicking the Contents button will present the Help index and clicking on a green underlined word will provide help relating to that subject.

EES commands are distributed among ten pull-down menus (Figure A.1-1). A brief summary of their functions follows.

Figure A.1-1: **Pull-down menus.**

The File menu provides commands for opening and saving work files and libraries, and for printing.

The **Edit** menu provides the editing commands to cut, copy, and paste information.

The **Search** menu provides **Find** and **Replace** commands for use in the Equations window.

The **Options** menu provides commands for: setting the guess values and bounds of variables; selecting the unit system, providing information on built-in and user-supplied library functions; and setting program preferences.

The **Calculate** menu contains the commands to check, format and solve the equation set. Commands to check the unit consistency of the equations and to update guess values are also provided.

The **Tables** menu contains commands to set up and alter the contents of the Parametric and Lookup Tables. The Parametric Table, which is similar to a spreadsheet, allows the equation set to be solved repeatedly while varying the values of one or more variables. The Lookup table holds user-supplied data which can be interpolated in one or two dimensions with internal functions and used in the solution of the equation set.

The **Plot** menu provides commands to prepare a new plot of data in the Parametric, Lookup, Array or Integral tables or to modify an existing plot. Curve-fitting capability and thermodynamic property plots are also provided.

The **Windows** menu provides a convenient method of bringing any of the EES windows to the front or to organize the windows.

The **Help** menu provides commands for accessing the online help documentation.

The **Heat Transfer** menu provides access to EES solutions to problems developed for this textbook.

A basic capability provided by EES is the solution of a set of non-linear algebraic equations. To demonstrate this capability, start EES and enter the simple example problem shown in Figure A.1-2 in the Equations window.

Figure A.1-2: **Example of equations.**

Text is entered in the same manner as for a word processor. Formatting rules are as follows:

1. Upper and lower case letters are not distinguished. EES will (optionally) change the case of all variables to match the manner in which they first appear.
2. Blank lines and spaces may be entered as desired since they are ignored.
3. Comments must enclosed within braces { } or within quote marks " ". Comments may span as many lines as needed. Comments within braces may be nested in which case only the outermost set of { } are in effect. Comments within quotes will also be displayed in the Formatted Equations window.
4. Variable names must start with a letter and consist of any keyboard characters except ( ) ‘ | * / + - ^ { } : " or ;. Array variables are identified with square braces around the array index or

indices, e.g., X[5]. String variables are identified by a $ as the last character. The maximum length of a variable name is 30 characters.

5. Equations can be as long as needed. Multiple equations may be entered on one line if they are separated by a semi-colon (;).
6. The caret symbol (^) or ** is used to indicate raising to a power.
7. The order in which the equations are entered does not matter.
8. The position of known and unknown variables in the equations does not matter.
9. The units of numerical constants can be entered by following the value with the unit designation in square brackets.

If you wish, you may view the equations in mathematical notation (Figure A.1-3) by selecting the Formatted Equations command from the Windows menu or from the Formatted Equations speed-button located below the menu bar.

Figure A.1-3: **Formatted equations window.**

Select the Solve command from the Calculate menu or press F2. A dialog window will appear indicating the progress of the solution. When the calculations are completed, the button will change from Abort to Continue (Figure A.1-4).



Figure A.1-4: **Dialog box for calculations.**

Click the Continue button. The solution to this equation set will then be displayed (Figure A.1-5).



Figure A.1-5: Solution window.

A.1.2 Using EES Effectively

In the subsequent section, EES will be used to solve an example heat transfer problem. The solution steps will attempt to show not only the nuts and bolts associated with operating the software but also some higher level concepts associated with using EES effectively.

EES has the ability to do two things that most software packages cannot do; it can solve equations regardless of the order that they are entered and it can assign not only values but also units to each variable in your problem. These are very powerful features; in our experience they also lead to misuse and confusion and so it is worth establishing firmly a best-practice methodology for using EES. We will attempt to adhere rigorously to this practice within the book and suggest that you do the same.

We are engineers and therefore most of the variables involved in our problems represent physical quantities rather than pure numbers; therefore they have units. The ability to work with problems and measurements that are expressed in a variety of units is a basic engineering skill. Used correctly, EES can essentially guarantee that you have not made any unit conversion errors; furthermore, EES can check for unit consistency among your equations which provides an additional check on your problem solution.

The first step in solving your problem should be to set your unit system to the basic SI unit system (Pa, N, m, kg, J, W, s, etc.). This is accomplished by selecting Unit System from the Options menu and selecting SI and the Kelvin, Pa, J, as shown in Figure A.1-6.



Figure A.1-6: Selecting the basic SI unit system.

The known information (the values provided by the problem statement) should be entered and immediately converted to these base SI units; we work in a country that is somewhat backwards with respect to its unit system and therefore it will often be necessary to convert from units such as inch, torr, atm, tons (or mass or of cooling) etc. to the more rational set of SI units. All of the variables used in the problem statement and solution should have the appropriate base SI unit; EES allows the units of each variable to be explicitly set, as discussed in the subsequent example.

The advantage of this technique is that these units are completely self-consistent. It is not necessary to do any unit conversions as you work the problem. Once the problem has been solved then you can convert the solution to whatever units are most appropriate. The other advantage of this technique is that EES can check that all of your equations are consistent with respect to the units that have been set for the variables that are involved. This capability provides a powerful check on your solution.

Finally, EES allows equations to be entered in any order; however, it is a good idea to enter the equations in a logical order. When possible, equations should be entered in a sequential manner so that an intermediate solution can be obtained after each equation is entered. This is similar to the way you would solve the problem by hand or using a conventional programming language (that is, wherever possible, each new equation should use only information that was stated or obtained from a previous solution). There are several reasons for doing this. First, it allows your program to be debugged as it is written; it is common for a student to enter all of the equations involved in a problem into EES before he or she attempts to solve the problem. When a solution cannot be obtained it is a frustrating process to debug the over-long program. Better to enter a single equation and see if it solves. When problems are encountered they can immediately be isolated to the last equation that was entered. This is good engineering practice; when something goes wrong with an experiment or device, an experienced engineer will immediately try to isolate the problem by testing each sub-system of the device.

There is another advantage to using the sequential approach discussed in the previous paragraph. EES solves a set of non-linear equations using an iterative approach; the approach starts from an initial point characterized by a set of "guess" values for each variable and iteratively improves the solution. The solution process is easier if the initial guess values for each variable are close to the final solution. By obtaining a series of intermediate solutions it is possible to always have a good set of guess values available; this allows the iterations that are required to go more smoothly.

A.1.3 A Heat Transfer Example Problem

In this section, a heat transfer problem is worked from start to finish illustrating some of the capabilities of the EES program. EES is particularly appropriate for this problem since the solution requires iterations that would tedious to do by hand.

A D = 0.5 mm diameter platinum hot wire (Figure A.1-7) that is L = 1.0 cm long is used to measure the velocity of air flowing in a duct.



Figure A.1-7: **A hot wire used to measure air velocity.**

The free-stream air temperature (T_∞) is 20°C and the air pressure is p=100 kPa. The resistance of the platinum wire (R) depends on its temperature (T) according to:

$$R = 0.30\left[\text{ohm/K}^{0.5}\right]\sqrt{T\text{-}270\ [\text{K}]} \qquad \text{(A.1-1)}$$

where R is in (ohm) and T is in (K). The hot wire is provided with a 0.1 amp current (i); the electrical dissipation is convected to the free stream so that the temperature of the wire depends on the heat transfer coefficient and therefore, on the velocity of the air. The measured voltage across the wire is related to the free stream velocity. Calculate and plot the voltage measured across the wire (V) for a range of free stream velocity (u_∞) between 20 m/s and 100 m/s. Assume that the wire temperature does not vary either axially or radially.

Start EES or select the New command from the File menu if you have already been using the program. A blank Equations window will appear. Since this problem will require use of the built-in properties for air, it is necessary to specify the unit system that will be used to obtain property information with the Unit System command in the Options menu (see Figure A.1-6). Note that the pressure will be expressed in Pa and energy units will be expressed in J with these choices. The first section of your EES program should contain the known information and accomplish the process of converting this information to the base SI unit system. Your code can be organized into sections that are identified with comments, as shown below:

```
"Appendix A: Heat Transfer Example"

"Input Information"
```

Because the text entered in the equation window was enclosed by quotes, it is not considered to be part of the problem. Rather, comments are used to make the code more readable. It is good engineering practice to provide comments with most every line of code that you write so that you can, at a glance, recall what the purpose of the code is. The other method for entering comments is to use the curly brackets, {}.

The known information is entered next, starting with the dimensions of the wire. The diameter of the wire is stated as 0.5 mm which is not in a base SI unit; therefore, the diameter must be converted from mm to m. This can be accomplished by dividing by 1000; however, you may not remember that mm and m are related by a factor of 1000. Also, the conversions between other units are not as easy to remember. Fortunately, EES has the built-in function convert which provides the conversion constant between any two dimensionally consistent units. The protocol for calling the convert function is:

Convert('From','To')

where 'From' and 'To' are unit designations. For example, to get the conversion factor required to convert mm to m would require Convert(mm,m). To view the units that are programmed in EES, select Unit Conversion Info from the Options menu (Figure A.1-8); the units are separated by dimension (the left scroll box).



Figure A.1-8: **Unit Conversion Information dialog box.**

Therefore, the wire dimensions are entered and converted according to:

```
D=0.5 [mm]*convert(mm,m)          "diameter of wire"
L=1.0 [cm]*convert(cm,m)          "length of wire"
```

Notice that the diameter is entered as 0.5 [mm] (the constant 0.5 has units mm) which is converted from mm to m using the convert function. Thus the variable D has units m. The units of the variables D and L must be explicitly set by the user. There are a number of ways to set the units. The easiest method is to select Variable Info from the Options menu (Figure A.1-9). The units for each variable can be entered in the column Units, as shown.



Figure A.1-9: **Variable Information dialog window.**

Select Check Units from the Calculate menu and EES will indicate that no unit problems are detected because the units of the constants, the variables, and the conversion factors are all consistent. To see this more clearly, select Formatted Equations from the Windows menu (Figure A.1-10) and notice that the units of the constants and conversion appropriately cancel. The Formatted Equations window provides the equations in a more readable format and these equations can be copied and pasted into a Word document in order to facilitate report writing.

Appendix A: Heat Transfer Example

Input Information

$$D = 0.5 \text{ [mm]} \cdot \left| 0.001 \cdot \frac{m}{mm} \right| \quad \text{diameter of wire}$$

$$L = 1 \text{ [cm]} \cdot \left| 0.01 \cdot \frac{m}{cm} \right| \quad \text{length of wire}$$

Figure A.1-10: **Formatted Equation window**

The remaining variables in the problem statement are entered:

```
i=0.5 [amp]                       "current"
```

```
T_infinity=converttemp(C,K,20 [C])    "free stream temperature"
u_infinity=20 [m/s]                   "free stream velocity"
P=100 [kPa]*convert(kPa,Pa)           "free stream pressure"
```

Notice that the process of converting the free stream temperature from °C to K does not involve a simple multiplication and therefore a special function (converttemp) is required. The calling protocol for converttemp is:

ConvertTemp('From','To',value)

where 'From' and 'To' are the temperature scales to the converted from and converted to, respectively, and value is the value of the temperature to be converted.

The units for each of the input parameters are entered in the Variable Information window.

At this point we are ready to begin to solve the problem, keeping in mind the strategy that the equations should be entered sequentially so that solutions can be obtained periodically. The solution to this problem begins with an energy balance on the wire. This is a steady-state problem since the voltage at a specified air velocity does not vary with time. The steady-state energy balance on the wire balances generation due to ohmic dissipation ($\dot{g}$) with heat transfer from the surface due to convection ($\dot{q}_{conv}$), as shown in Figure A.1-7:

$$\dot{g} = \dot{q}_{conv} \qquad \text{(A.1-2)}$$

where $\dot{g}$ is the rate of thermal energy generation by ohmic dissipation and $\dot{q}_{conv}$ is the convective heat transfer rate from the wire surface. The energy generation rate can be described by:

$$\dot{g} = i^2 R \qquad \text{(A.1-3)}$$

The resistance of the platinum wire is a known function of temperature.

$$R = \frac{0.30\,\left[\text{ohm/K}^{0.5}\right]}{\sqrt{T\text{-}270\ [\text{K}]}} \qquad \text{(A.1-4)}$$

The product of the current and the resistance is the voltage:

$$V = i\,R \qquad \text{(A.1-5)}$$

We do not know the temperature of the wire at this point; this problem is inherently iterative in that it is not possible to progress linearly from a starting point through a series of equations to an ending point. Most heat transfer problems are iterative. In fact, most of the problems that you will encounter in your engineering career, outside of text books, will be iterative; textbook problems are often not iterative because they were designed to be solved using a pencil, paper, and calculator; the real world is not this considerate.

The best thing to do at this point is assume a reasonable value for temperature and proceed to solve the problem. When you are done, you can remove the assumption to obtain a final solution. This is analogous to the pencil and paper approach of iteration but much easier because the actual iteration process is done automatically. A good assumption for the wire temperature to start with is somewhat higher than the free stream air temperature (for example, T_infinity+20).

```
"Solution"
T=T_infinity+20 [K]
        "initial guess for the wire temperature - note that this will be removed to complete the iteration"
```

The temperature of the wire is used to compute the resistance of the wire with Eq. (A.1-4), the generation with Eq. (A.1-3), and the voltage with Eq. (A.1-5):

```
R=0.3 [ohm/K^0.5]*sqrt(T-270[K])     "wire resistance"
g_dot=i^2*R                          "generation in the wire"
V=i*R                                "voltage"
```

The units for the variables that have been introduced (i.e., T, R, g_dot, and V) should be set in the Variable Information window. Note that it is possible at this point to check the units and obtain a solution. In short, the validity of the solution to this point can be established and if problems are found then they can be addressed quickly because there are only 3 lines of code that might be the source of any errors.

The convective heat transfer rate is given by Newton's Law.

$$\dot{q}_{conv} = \overline{h}\, A_s \left(T - T_\infty \right) \tag{A.1-6}$$

where $\overline{h}$ is the average heat transfer coefficient and A_s is the surface area of the wire:

$$A_s = \pi\, D\, L \tag{A.1-7}$$

The complication arising in this problem, as in many heat transfer problems, is that the heat transfer coefficient is not known a-priori. Consequently it is necessary to estimate $\overline{h}$ using the methods that are discussed in the convection chapters. For this geometry, the non-dimensional heat transfer coefficient (i.e, the Nusselt number, $\overline{Nu}$) is a function of the Reynolds number, *Re*, and Prandtl number (*Pr*).

$$\overline{Nu} = \frac{\overline{h}\, D}{k} \tag{A.1-8}$$

$$Re = \frac{u_\infty\, D\, \rho}{\mu} \tag{A.1-9}$$

where k, ρ, and μ are the thermal conductivity, density, and viscosity of air evaluated at the film temperature. The Prandtl number is also a property of air that should be evaluated at the film temperature. The film temperature, T_f, is the average of T_∞ and T:

$$T_f = \frac{T + T_\infty}{2} \tag{A.1-10}$$

The film temperature and surface area can be calculated using Eqs. (A.1-10) and (A.1-7):

```
T_f=(T+T_infinity)/2                    "film temperature"
A_s=pi*D*L                              "surface area"
```

Note that the variable pi is a built-in constant that contains the value of π. The required properties of air (and many other substances) are built into EES. To access these properties select Function Info from the Options menu and then select Fluid properties (Figure A.1-11).



Figure A.1-11: **Function Information window showing the Fluid property functions**

The left scroll box includes the various properties that can be obtained and the right scroll box includes the substances that are available. Scroll to the property Conductivity in the left scroll box and the substance Air in the right scroll box. Select the Paste button and the function will be pasted into the Equations window, with a little modification the equation becomes:

```
k=Conductivity(Air,T=T) "conductivity of air"
```

The function Conductivity requires two calling parameters; the first indicates the substance (Air) and the second parameter (T=T) actually includes two pieces of information. The quantity on the left side of the equal sign tells EES that the independent variable used to specify conductivity will be temperature (T) while the right side of the equal sign tells EES the value of the independent variable (equal to the variable T in the problem).

The units for the conductivity returned by the function Conductivity will be consistent with the unit system that you set (in Figure A.1-6). Therefore, you should set the units of the variable k to W/m-K. Along the same line, temperature that you supply will be interpreted as having units K because this is consistent with the unit system specified. The viscosity, density, and Prandtl number are calculated using the same process:

```
mu=Viscosity(Air,T=T)            "Viscosity of air"
rho=Density(Air,T=T,P=P)         "Density of air"
Pr=Prandtl(Air,T=T)              "Prandtl number of air"
```

Note that density depends on both temperature and pressure. The substance Air in EES corresponds to the ideal gas, Air. Many other, more complex and non-ideal substances are included in EES; for example, Air_ha corresponds to more accurate properties of air (including its saturation properties). For non-ideal substances, two independent properties must be used to specify any dependent property. For example, to obtain the conductivity of air using the substance Air_ha would require:

```
k=Conductivity(Air_ha,T=T,P=P) "conductivity of air using the more accurate Air_ha substance"
```

The conductivity could be specified with a different set of independent properties (for example enthalpy and pressure rather than temperature and pressure). In this case, the 1st argument would change from T= to h=:

```
k=Conductivity(Air_ha,h=100000 [J/kg],P=P)
        "conductivity of air using the more accurate Air_ha substance based on enthalpy and pressure"
```

For this problem, the ideal gas model of air is sufficient and therefore the correct equation is:

```
k=Conductivity(Air,T=T) "conductivity of air"
```

The Reynolds number can be computed using Eq. (A.1-9):

```
Re=u_infinity*D*rho/mu "Reynolds number"
```

Again, it is useful to set the units of all of the variables and check the units. Obtain a solution using the Solve selection from the Calculate menu and verify that everything looks reasonable.

A substantial portion of this and other heat transfer books is dedicated to the relationships between Nusselt number and Reynolds number for various flow situations. Many of these relationships are programmed in EES and can be accessed by selecting Function Info from the Options menu and then selecting the lower right scroll bar. Scroll down to Convection and then select External Flow from the list. There are a number of correlations that can be accessed using the lower scroll bar (Figure A.1-12).



Figure A.1-12: **Function Information showing the External Flow Convection functions that are integrated with EES.**

For this problem we will use the relationship between the Nusselt and Reynold's number that is provided below, rather than a built-in function:

$$\overline{Nu} = 0.3 + 0.62\sqrt{Re}\,Pr^{1/3}\frac{\left(1+\left(Re/282000\right)^{5/8}\right)^{4/5}}{\left(1+\left(0.4/Pr\right)^{2/3}\right)^{1/4}} \qquad \text{(A.1-11)}$$

which is implemented in EES:

```
Nusselt=0.3+0.62*sqrt(Re)*Pr^(1/3)*(1+(Re/282000)^(5/8))^(4/5)/(1+(0.4/Pr)^(2/3))^(1/4)
          "convection relation"
```

A complex equation such as Eq. (A.1-11) should be checked in the Formatted Equations window (Figure A.1-13) where the structure of the equation is clearer and errors can be picked out.

$$\text{Nusselt} = 0.3 + 0.62 \cdot \sqrt{\text{Re}} \cdot \text{Pr}^{(1/3)} \cdot \left[\frac{\left(1 + \left[\frac{\text{Re}}{282000}\right]^{(5/8)}\right)^{(4/5)}}{\left(1 + \left[\frac{0.4}{\text{Pr}}\right]^{(2/3)}\right)^{(1/4)}}\right] \qquad \text{convection relation}$$

Figure A.1-13: **Nusselt number equation in the Formatted Equations window.**

The Nusselt number is used to calculate the heat transfer coefficient using Eq. (A.1-8) and the convection heat transfer rate using Eq. (A.1-6):

```
Nusselt=h_bar*D/k                    "heat transfer coefficient"
q_dot_conv=h_bar*A_s*(T-T_infinity)  "convection heat transfer rate"
```

At this point it is possible to calculate the generation rate, $\dot{g}$=15.7 W, and convection heat transfer rate, $\dot{q}_{conv}$=0.21 W. These quantities are supposed to be equal according to Eq. (A.1-2); because we selected a wire temperature arbitrarily it would have been extraordinary luck it they were equal. If this were a pencil and paper solution we would re-calculate T based on the convection rate and iterate through the set of equations again (and repeat until $\dot{g} = \dot{q}_{conv}$).

EES removes the drudgery of manual iteration. The solution that you currently have, although wrong, is still a very good starting point for EES' iteration process. Therefore, select Update Guesses from the Calculate menu so that the guess value for each variable is set to its current value. The guess values for each variable can be accessed in the Variable Information window.

Remove the guess value that you used for temperature; this can be done quickly by highlighting the line of code, right-clicking, and selecting Comment {}. Now add the energy balance, Eq. (A.1-2). Your code should look like:

```
"Appendix A: Heat Transfer Example"

"Input Information"
D=0.5 [mm]*convert(mm,m)                "diameter of wire"
L=1.0 [cm]*convert(cm,m)                "length of wire"
i=0.5 [amp]                             "current"
T_infinity=converttemp(C,K,20)          "free stream temperature"
u_infinity=20 [m/s]                     "free stream velocity"
P=100 [kPa]*convert(kPa,Pa)             "free stream pressure"

"Solution"
{T=T_infinity+20 [K]
        "initial guess for the wire temperature - note that this will be removed to complete the iteration"}
R=0.3 [ohm/K^0.5]*sqrt(T-270[K])        "wire resistance"
g_dot=i^2*R                             "generation in the wire"
V=i*R                                   "voltage"
T_f=(T+T_infinity)/2                    "film temperature"
A_s=pi*D*L                              "surface area"
k=Conductivity(Air,T=T)                 "conductivity of air"
mu=Viscosity(Air,T=T)                   "Viscosity of air"
rho=Density(Air,T=T,P=P)                "Density of air"
Pr=Prandtl(Air,T=T)                     "Prandtl number of air"
Re=u_infinity*D*rho/mu                  "Reynolds number"
Nusselt=0.3+0.62*sqrt(Re)*Pr^(1/3)*(1+(Re/282000)^(5/8))^(4/5)/(1+(0.4/Pr)^(2/3))^(1/4)
        "convection relation"
Nusselt=h_bar*D/k                       "heat transfer coefficient"
q_dot_conv=h_bar*A_s*(T-T_infinity)     "convection heat transfer rate"
g_dot=q_dot_conv                          "energy balance"
```

The solution window is shown in Figure A.1-14 and indicates that the voltage measured by the hot wire when the free stream velocity is 20 m/s will be 1.458 V.



Figure A.1-14: **Solution window.**

One of the most useful features of EES is its ability to provide parametric studies. In this problem, for example, we were asked to determine the voltage (V) as a function of air velocity, (u_∞) for values ranging between 20 and 100 m/s. The solution that is shown in Figure A.1-14 is for u_∞ = 20 m/s. A series of calculations at other velocities can be obtained automatically and plotted in EES.

Select the New Table command in the Tables menu. A dialog will be displayed listing the variables appearing in the Equations window (Figure A.1-15). In this case, we will construct a table containing the variables u_infinity, V, and T. Click on these variables in the list on the left.



Figure A.1-15: **New Table dialog.**

Click the Add button to move the selected variables to the list on the right. As a short cut, you can double-click on a variable name in the list on the left to directly move it to the list on the right. Note that you can set the number of runs (rows) in the Parametric table and give the table a name. These changes can be made now or at a later time. Click the OK button to create the table.

The Parametric Table works much like a spreadsheet. You can type numbers directly into the cells; numbers that you enter are shown in black and produce the same effect as if you set the variable to that value with an equation in the Equations window. We will be entering values for u_∞ into the table. For that reason, it will be necessary to remove it from the Equations window (otherwise you will over-constrain the problem by defining the variable twice). There are several ways to remove the variable but the simplest is to enclose the equation u_infinity=20 [m/s] in comment brackets { }; this can be done by highlighting the equation, right-clicking and selecting Comment from the popup menu. Now it is necessary to enter values of u_infinity in the Parametric Table. You could simply type values into the table cell, but it is easier to have EES enter the values automatically. Automatic entry is initiated by clicking on the triangular icon at the upper right of the header cell for the u_infinity column in the table. This action will bring up the Alter values dialog. Enter 20 for the first value and 100 for the last value (Figure A.1-16). EES will automatically fill in the values for all rows in the Parametric table when the dialog is dismissed by clicking the OK button.



Figure A.1-16: **Alter values dialog.**

At this point, the Parametric Table should appear as shown in Figure A.1-17.



1..10	T [K]	u_∞ [m/s]	V [V]
Run 1		20	
Run 2		28.89	
Run 3		37.78	
Run 4		46.67	
Run 5		55.56	
Run 6		64.44	
Run 7		73.33	
Run 8		82.22	
Run 9		91.11	
Run 10		100	

Figure A.1-17: **Parametric table.**

Now, select Solve Table from the Calculate menu or press F3. The Solve Table dialog window will appear (Figure A.1-18) allowing you to choose the runs for which calculations will be done.



Figure A.1-18: **Solve table dialog.**

Note that the Update Guess Values control is selected in Figure. A.1-18; this indicates that the solution for the last run (or row) will provide guess values for the subsequent run (or row). Click the OK button. A status window will be displayed, indicating the progress of the solution. When the calculations are completed, the values calculated for V and T are entered into the table (Figure A.1-19). The values calculated by EES will be displayed in blue, bold or italic type depending on the setting made in the Screen Display tab of the Preferences dialog window in the Options menu.

Table 1

1..10	T [K]	u_{∞} [m/s]	V [V]
Run 1	364.5	20	1.458
Run 2	346.5	28.89	1.312
Run 3	336.6	37.78	1.224
Run 4	330.3	46.67	1.164
Run 5	325.8	55.56	1.12
Run 6	322.5	64.44	1.087
Run 7	319.9	73.33	1.059
Run 8	317.8	82.22	1.037
Run 9	316.1	91.11	1.018
Run 10	314.7	100	1.002

Figure A.1-19: **Parametric table after solving.**

The relationship between variables such as V and u_infinity is apparent by examining Figure A.1-19 but it can be seen more clearly with a plot. Select New Plot Window from the Plot menu. The New Plot Window dialog window shown in Figure A.1-20 will appear. Choose u_infinity to be the x-axis by clicking on it in the X-Axis list and click on V in the Y-Axis list. You may wish to adjust the scale limits or add grid lines. The spline fit option will draw a smooth line through the data points, which is useful when there are few points as in this case. When you click the OK button, the plot will be constructed and the plot window will appear as shown in Figure A.1-21.



Figure A.1-20: **New Plot Window dialog.**



Figure A.1-21: **Plot window.**

Once created, there are a variety of ways in which the appearance of the plot can be changed. Double-click the mouse in the plot rectangle or on the plot axis to see some of these options.

A.1.4 Using the Heat Transfer Menu

A number of problems from this textbook have been solved using EES. These examples are accessible from the Heat Transfer menu to the right of the Help menu. As an example, select Chapter 4 from the Heat Transfer menu. A dialog window will appear listing the problems in Chapter 4.

At this point, you should explore. Try whatever you wish. You can't hurt anything. The online help (invoked by pressing F1) will provide details for the EES commands. EES is a powerful tool that you will find very useful in your studies.